Spreadsheet Changes for WBT 4.2.1

February 15, 2013

1. Water Vaporized During Simmer

Sheets: Test-1, Test-2, Test-3

Cell: AP27

Old formula:	AP27=(AL12-AP12)+(AL13-AP13)+(AL14-AP14)+(AL15-AP15)
New formula:	AP27=(AL12-AP12)+(AP13-AL13)+(AP14-AL14)+(AP15-AL15)

The water vaporized is the initial water mass minus the final water mass. The old formula is correct for Pot 1, but not for Pot 2, Pot 3, and Pot 4. The new formula is corrected to match the equation:

$$w_{sv} = \sum_{j=1}^{4} \left(Pj_{si} - Pj_{sf}\right)$$

2. Column widths

Sheets: Test-1, Test-2, Test-3

Cells: All

The column widths were set to be the same for all three test sheets.

Spreadsheet Changes for WBT 4.2.0

January 20, 2013

1. Duct Temperature (C)

Sheet: General Information

Cell:C23

The duct temperature was added as an input for the emission measurements. The emission calculations relied on the ambient temperature, which can be considerably lower than the duct temperature in the collection hood. The duct temperature was added as an input to improve the accuracy of the emission calculations. The duct temperature represents the average duct temperature over the duration of the test. The duct temperature was also added as an input on the Test Entry sheet.

2. Exhaust Carbon Concentration Cold Start (ppm)

Sheets: Test-1,Test-2,Test-3

Cell: W42

Old formula:

W42=((AA18-$N21)+(AA19-$N22)+(AA20-$N23)/15/('General Information'!$C$20/0.008314/(E15+273.15)))

New formula:

W42=((AA18-$N21)+(AA19-$N22)+(AA20-$N23)/15/('General Information'!$C$20/0.008314/('General Information'!C23+273.15)))

The formula was changed to reference the duct temperature ('General Information'!C23) instead of the ambient temperature (E15).

3. Exhaust Carbon Concentration Hot Start (ppm)

Sheets: Test-1,Test-2,Test-3

Cell: AA42

Old formula:

AA42=((AI18-$N21)+(AI19-$N22)+(AI20-$N23)/15/('General Information'!$C$20/0.008314/(E15+273.15)))

New formula:

AA42=((AI18-$N21)+(AI19-$N22)+(AI20-$N23)/15/('General Information'!$C$20/0.008314/('General Information'!C23+273.15)))

The formula was changed to reference the duct temperature ('General Information'!C23) instead of the ambient temperature (E15).

4. Exhaust Carbon Concentration Simmer (ppm)

Sheets: Test-1,Test-2,Test-3

Cell: AE42

Old formula:

AE42=((AP18-$N21)+(AP19-$N22)+(AP20-$N23)/15/('General Information'!$C$20/0.008314/(E15+273.15)))

New formula:

AE42=((AP18-$N21)+(AP19-$N22)+(AP20-$N23)/15/('General Information'!$C$20/0.008314/('General Information'!C23+273.15)))

The formula was changed to reference the duct temperature ('General Information'!C23) instead of the ambient temperature (E15).

5. Total Carbon in Exhaust Cold Start (g/m^3)

Sheets: Test-1,Test-2,Test-3

Cell: W43

Old formula:

W43='General Information'!$C$20/0.008314/($E$15+273.15)*W42*0.000001*12

New formula:

W43='General Information'!$C$20/0.008314/('General Information'!C23+273.15)*W42*0.000001*12

The formula was changed to reference the duct temperature ('General Information'!C23) instead of the ambient temperature (E15).

6. Total Carbon in Exhaust Hot Start (g/m^3)

Sheets: Test-1,Test-2,Test-3

Cell: AA43

Old formula:

AA43='General Information'!$C$20/0.008314/($E$15+273.15)*AA42*0.000001*12

New formula:

AA43='General Information'!$C$20/0.008314/('General Information'!C23+273.15)*AA42*0.000001*12

The formula was changed to reference the duct temperature ('General Information'!C23) instead of the ambient temperature (E15).

7. Total Carbon in Exhaust Simmer (g/m^3)

Sheets: Test-1,Test-2,Test-3

Cell: AE43

Old formula:

AE43='General Information'!$C$20/0.008314/($E$15+273.15)*AE42*0.000001*12

New formula:

AE43='General Information'!$C$20/0.008314/('General Information'!C23+273.15)*AE42*0.000001*12

The formula was changed to reference the duct temperature ('General Information'!C23) instead of the ambient temperature (E15).

8. Char Carbon Content (%)

Sheet: General Information

Cell:L12

L12='Calorific values'!M8

The char carbon content is assumed to equal the fuel carbon content of charcoal fuel. The char carbon content was added in order to calculate the Hood Carbon Balance metric and Total Emission explained below. To make room on the General Information sheet, the Check Box for Measured Calorific Values was moved down one cell from H14 to H15, which automatically changes all references to that cell.

9. Hood Carbon Balance Cold Start (%)

Sheets: Test-1,Test-2,Test-3

Cell: W45

Old formula:

W45=IF(W24=0,0,W44*(1-$E$18)/(W24))

New formula:

W45=IF(W24=0,0,W44/((1-$E$18)*(W24)-W25*charFracC/'General Information'!L23))

The carbon balance is the fraction of total carbon emissions collected by the emissions hood. It is the ratio of total carbon measured in the emissions to total carbon measured in the consumed fuel.

$$\text{Carbon Balance} = \frac{\text{carbon emission collected}}{\text{total carbon emitted}}$$

$$= \frac{(\text{dry fuel collected in emissions}) * (\text{carbon fraction of fuel})}{(\text{dry fuel consumed}) * (\text{carbon fracion of fuel})} = \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}}$$

The old formula is:

$$\text{CB} = \frac{f_{ce} \cdot (1 - \text{MC})}{f_{cm}} \neq \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}}$$

CB=carbon balance (fraction)

f_{ce}=dry fuel consumed estimated from emissions (g)

MC=moisture content of fuel dry basis (fraction)

f_{cm}=moist fuel consumed (g)

The new formula is:

$$\text{CB} = \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}} = \frac{f_{ce}}{f_{cm} \cdot (1 - \text{MC}) - \frac{\triangle\, c_c \cdot \text{charFracC}}{\text{fuelFracC}}}$$

CB=carbon balance (fraction)

f_{ce}=dry fuel consumed estimated from emissions (g)

f_{cm}=moist fuel consumed (g)

MC=moisture content of fuel dry basis (fraction)

$\triangle\, c_c$=net change in char during test (g)

charFracC=carbon fraction of char (fraction)

fuelFracC=carbon fraction of fuel (fraction)

The dry fuel consumed is the moist fuel consumed minus the mass of moisture in the fuel minus the mass of fuel remaining in the form of char. The value of CB is a decimal fraction but the cell is formatted as a percent.

10. Hood Carbon Balance Hot Start (%)

Sheets: Test-1,Test-2,Test-3

Cell: AA45

Old formula:

AA45=IF(AA24=0,0,AA44*(1-$E$18)/(AA24))

New formula:

AA45=IF(AA24=0,0,AA44/((1-$E$18)*(AA24)-AA25*charFracC/'General Information'!L23))

The carbon balance is the fraction of total carbon emissions collected by the emissions hood. It is the ratio of total carbon measured in the emissions to total carbon measured in the consumed fuel.

$$\text{Carbon Balance} = \frac{\text{carbon emission collected}}{\text{total carbon emitted}}$$

$$= \frac{(\text{dry fuel collected in emissions})*(\text{carbon fraction of fuel})}{(\text{dry fuel consumed})*(\text{carbon fracion of fuel})} = \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}}$$

The old formula is:

$$\text{CB} = \frac{f_{he} \cdot (1 - \text{MC})}{f_{hm}} \neq \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}}$$

CB=carbon balance (fraction)

f_{he}=dry fuel consumed estimated from emissions (g)

MC=moisture content of fuel dry basis (fraction)

f_{hm}=moist fuel consumed (g)

The new formula is:

$$\text{CB} = \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}} = \frac{f_{he}}{f_{hm} \cdot (1 - \text{MC}) - \frac{\Delta\, c_h \cdot \text{charFracC}}{\text{fuelFracC}}}$$

CB=carbon balance (fraction)

f_{he}=dry fuel consumed estimated from emissions (g)

f_{hm}=moist fuel consumed (g)

MC=moisture content of fuel dry basis (fraction)

Δc_h=net change in char during test (g)

charFracC=carbon fraction of char (fraction)

$fuelFracC$=carbon fraction of fuel (fraction)

The dry fuel consumed is the moist fuel consumed minus the mass of moisture in the fuel minus the mass of fuel remaining in the form of char. The value of CB is a decimal fraction but the cell is formatted as a percent.

11. Hood Carbon Balance Simmer (%)

Sheets: Test-1,Test-2,Test-3

Cell: AE45

Old formula:

AE45=IF(AP24=0,0,AE44*(1-$E$18)/(AP24))

New formula:

AE45=IF(AP24=0,0,AE44/((1-$E$18)*(AP24)-AP25*charFracC/'General Information'!L23))

The carbon balance is the fraction of total carbon emissions collected by the emissions hood. It is the ratio of total carbon measured in the emissions to total carbon measured in the consumed fuel.

$$\text{Carbon Balance} = \frac{\text{carbon emission collected}}{\text{total carbon emitted}}$$

$$= \frac{(\text{dry fuel collected in emissions})*(\text{carbon fraction of fuel})}{(\text{dry fuel consumed})*(\text{carbon fracion of fuel})} = \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}}$$

The old formula is:

$$\mathrm{CB} = \frac{\mathrm{f_{se}} \cdot (1 - \mathrm{MC})}{\mathrm{f_{sm}}} \neq \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}}$$

CB=carbon balance (fraction)

f_{se}=dry fuel consumed estimated from emissions (g)

MC=moisture content of fuel dry basis (fraction)

f_{sm}=moist fuel consumed (g)

The new formula is:

$$\mathrm{CB} = \frac{\text{dry fuel collected in emissions}}{\text{dry fuel consumed}} = \frac{\mathrm{f_{se}}}{\mathrm{f_{sm}} \cdot (1 - \mathrm{MC}) - \frac{\Delta\, \mathrm{c_s} \cdot \mathrm{charFracC}}{\mathrm{fuelFracC}}}$$

CB=carbon balance (fraction)

f_{se}=dry fuel consumed estimated from emissions (g)

f_{sm}=moist fuel consumed (g)

MC=moisture content of fuel dry basis (fraction)

$\Delta\, c_s$=net change in char during test (g)

charFracC=carbon fraction of char (fraction)

fuelFracC=carbon fraction of fuel (fraction)

The dry fuel consumed is the moist fuel consumed minus the mass of moisture in the fuel minus the mass of fuel remaining in the form of char. The value of CB is a decimal fraction but the cell is formatted as a percent.

12. CO2 mass emission factor for Simmer (g_{CO2}/kg_{fuel})

Sheets: Test-1,Test-2,Test-3

Cell: AE51

Old formula:

AE51=IF(AE42>0,((AQ18-$N21)*44/12)/(AE$42/fuelFracC)*1000,0)

New formula:

AE51=IF(AE42>0,((AP18-$N21)*44/12)/(AE$42/fuelFracC)*1000,0)

The formula incorrectly referenced cell AQ18, and was corrected to reference cell AP18 instead.

13. Total Emissions

Sheets: Test-1,Test-2,Test-3

Cells: T54:AH59

A block of cells was inserted here to calculate and report Total Emissions, which are used to calculate the IWA performance metrics.

14. Total Emission CO_2 Cold Start (g)

Sheets: Test-1,Test-2,Test-3

Cell: W55

Formula: W55=W51*(W24*(1-$E$18)-W25*charFracC/fuelFracC)/1000

$$m_{CO2,c} = EF_{CO2,c} \cdot \left(f_{cm} \cdot (1 - MC) - \Delta c_c * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{CO2,c} =$ total emission (g)

$EF_{CO2,c}$ = emission factor ($g_{emission}/kg_{fuel}$)

f_{cm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_c = net change in char during test (g)

$charFracC$ = char carbon content (decimal fraction)

$fuelFracC$ = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

15. Total Emission CO_2 Hot Start (g)

Sheets: Test-1,Test-2,Test-3

Cell: AA55

Formula: AA55=AA51*(AA24*(1-$E$18)-AA25*charFracC/fuelFracC)/1000

$$m_{CO2,h} = EF_{CO2,h} \cdot \left(f_{hm} \cdot (1 - MC) - \Delta c_h * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{CO2,h} =$ total emission (g)

$EF_{CO2,h}$= emission factor ($g_{emission}/kg_{fuel}$)

f_{hm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_h = net change in char during test (g)

charFracC = char carbon content (decimal fraction)

fuelFracC = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

16. Total Emission CO_2 Simmer (g)

Sheets: Test-1,Test-2,Test-3

Cell: AE55

Formula: AE55=AE51*(AP24*(1-$E$18)-AP25*charFracC/fuelFracC)/1000

$$m_{CO2,s} = EF_{CO2,s} \cdot \left(f_{sm} \cdot (1 - MC) - \Delta c_s * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{CO2,s} =$ total emission (g)

$EF_{CO2,s}$ = emission factor ($g_{emission}/kg_{fuel}$)

f_{sm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_s = net change in char during test (g)

$charFracC$ = char carbon content (decimal fraction)

$fuelFracC$ = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

17. Total Emission CO Cold Start (g)

Sheets: Test-1,Test-2,Test-3

Cell: W56

Formula: W56 =W52*(W24*(1-$E$18)-W25*charFracC/fuelFracC)/1000

$$m_{CO,c} = EF_{CO,c} \cdot \left(f_{cm} \cdot (1 - MC) - \Delta c_c * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{CO,c}$ = total emission (g)

$EF_{CO,c}$ = emission factor ($g_{emission}/kg_{fuel}$)

f_{cm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_c = net change in char during test (g)

$charFracC$ = char carbon content (decimal fraction)

$fuelFracC$ = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

18. Total Emission CO Hot Start (g)

Sheets: Test-1,Test-2,Test-3

Cell: AA56

Formula: AA56=AA52*(AA24*(1-$E$18)-AA25*charFracC/fuelFracC)/1000

$$m_{CO,h} = EF_{CO,h} \cdot \left(f_{hm} \cdot (1 - MC) - \Delta c_h * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{CO,h} =$ total emission (g)

$EF_{CO,h}$= emission factor ($g_{emission}/kg_{fuel}$)

f_{hm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_h = net change in char during test (g)

charFracC = char carbon content (decimal fraction)

fuelFracC = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

19. Total Emission CO Simmer (g)

Sheets: Test-1,Test-2,Test-3

Cell: AE56

Formula: AE56 =AE52*(AP24*(1-$E$18)-AP25*charFracC/fuelFracC)/1000

$$m_{CO,s} = EF_{CO,s} \cdot \left(f_{sm} \cdot (1 - MC) - \Delta c_s * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{CO,s} =$ total emission (g)

$EF_{CO,s}$ = emission factor ($g_{emission}/kg_{fuel}$)

f_{sm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_s = net change in char during test (g)

$charFracC$ = char carbon content (decimal fraction)

$fuelFracC$ = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

20. Total Emission PM Cold Start (g)

Sheets: Test-1,Test-2,Test-3

Cell: W57

Formula: W57=W53*(W24*(1-$E$18)-W25*charFracC/fuelFracC)/1000

$$m_{PM,c} = EF_{PM,c} \cdot \left(f_{cm} \cdot (1 - MC) - \Delta c_c * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{PM,c} =$ total emission (g)

$EF_{PM,c}$ = emission factor ($g_{emission}/kg_{fuel}$)

f_{cm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_c = net change in char during test (g)

$\text{charFracC} = \text{char carbon content (decimal fraction)}$

$\text{fuelFracC} = \text{fuel carbon content (decimal fraction)}$

$\frac{1}{1000} = \text{conversion factor } \left(\frac{\text{kg}}{\text{g}}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

21. Total Emission PM Hot Start (g)

Sheets: Test-1,Test-2,Test-3

Cell: AA57

Formula: AA57 =AA53*(AA24*(1-$E$18)-AA25*charFracC/fuelFracC)/1000

$$m_{PM,h} = EF_{PM,h} \cdot \left(f_{hm} \cdot (1 - MC) - \Delta c_h * \frac{\text{charFracC}}{\text{fuelFracC}}\right) * \frac{1}{1000}$$

$m_{PM,h} = \text{ total emission (g)}$

$EF_{PM,h}$= emission factor ($g_{emission}/kg_{fuel}$)

$f_{hm} = \text{moist fuel consumed (g)}$

$MC = \text{fuel moisture content (decimal fraction)}$

$\Delta c_h = \text{net change in char during test (g)}$

$\text{charFracC} = \text{char carbon content (decimal fraction)}$

$\text{fuelFracC} = \text{fuel carbon content (decimal fraction)}$

$\frac{1}{1000} = \text{conversion factor } \left(\frac{\text{kg}}{\text{g}}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

22. Total Emission PM Simmer (g)

Sheets: Test-1,Test-2,Test-3

Cell: AE57

Formula: AE57 =AE53*(AP24*(1-$E$18)-AP25*charFracC/fuelFracC)/1000

$$m_{PM,s} = EF_{PM,s} \cdot \left(f_{sm} \cdot (1 - MC) - \Delta c_s * \frac{charFracC}{fuelFracC}\right) * \frac{1}{1000}$$

$m_{PM,s} =$ total emission (g)

$EF_{PM,s}$ = emission factor ($g_{emission}/kg_{fuel}$)

f_{sm} = moist fuel consumed (g)

MC = fuel moisture content (decimal fraction)

Δc_s = net change in char during test (g)

$charFracC$ = char carbon content (decimal fraction)

$fuelFracC$ = fuel carbon content (decimal fraction)

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The total emission [g] is the emission factor [$g_{emission}/kg_{dryfuel}$] multiplied by the dry fuel consumed [$kg_{dryfuel}$]. The dry fuel consumed is the moist fuel consumed minus the weight of moisture in the fuel minus the amount of fuel that remains in the form of char.

23. Cold Start CO2 Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: W47

Old formula: W47=W51*W$33/1000

New formula: W47=W55/W28*1000

The old formula is expressed as: $E_{CO2,c} = EF_{CO2,c} \cdot SC_c \cdot \frac{1}{1000}$

$E_{CO2,c}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$\mathrm{EF_{CO2,c}}$ = emission factor $\left(\frac{\mathrm{g_{emission}}}{\mathrm{kg_{dry\_fuel}}}\right)$

$\mathrm{SC_c}$ = specific fuel consumption $\left(\frac{\mathrm{g_{eq\_dry\_fuel}}}{\mathrm{L_{water}}}\right)$

$\frac{1}{1000}$ = conversion factor $\left(\frac{\mathrm{kg}}{\mathrm{g}}\right)$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$\mathrm{E_{CO2,c}} = \frac{\mathrm{m_{CO2,c}}}{\mathrm{w_{cr}}} \cdot 1000$$

$\mathrm{E_{CO2,c}}$ = emission per water boiled $\left(\frac{\mathrm{g_{emission}}}{\mathrm{L_{water}}}\right)$

$\mathrm{m_{CO2,c}}$ = total emission (g)

$\mathrm{w_{cr}}$ = water boiled (g)

1000 = conversion factor $\left(\frac{\mathrm{g_{water}}}{\mathrm{L_{water}}}\right)$

24. Hot Start CO2 Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: AA47

Old formula: AA47=AA51*AA$33/1000

New formula: AA47=AA55/AA28*1000

The old formula is expressed as: $\mathrm{E_{CO2,h}} = \mathrm{EF_{CO2,h}} \cdot \mathrm{SC_h} \cdot \frac{1}{1000}$

$\mathrm{E_{CO2,h}}$ = emission per water boiled $\left(\frac{\mathrm{g_{emission}}}{\mathrm{L_{water}}}\right)$

$\mathrm{EF_{CO2,h}}$ = emission factor $\left(\frac{\mathrm{g_{emission}}}{\mathrm{kg_{dry\_fuel}}}\right)$

$\mathrm{SC_h}$ = specific fuel consumption $\left(\frac{\mathrm{g_{eq\_dry\_fuel}}}{\mathrm{L_{water}}}\right)$

$$\frac{1}{1000} = \text{conversion factor} \left(\frac{\text{kg}}{\text{g}}\right)$$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{CO2,h} = \frac{m_{CO2,h}}{w_{hr}} \cdot 1000$$

$$E_{CO2,h} = \text{emission per water boiled} \left(\frac{g_{emission}}{L_{water}}\right)$$

$$m_{CO2,h} = \text{total emission (g)}$$

$$w_{hr} = \text{water boiled (g)}$$

$$1000 = \text{conversion factor} \left(\frac{g_{water}}{L_{water}}\right)$$

25. Simmer CO2 Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: AE47

Old formula: AE47=AE51*AP$32/1000

New formula: AE47=AE55/AP28*1000

The old formula is expressed as: $E_{CO2,s} = EF_{CO2,s} \cdot SC_s \cdot \frac{1}{1000}$

$$E_{CO2,s} = \text{emission per water boiled} \left(\frac{g_{emission}}{L_{water}}\right)$$

$$EF_{CO2,s} = \text{emission factor} \left(\frac{g_{emission}}{kg_{dry\_fuel}}\right)$$

$$SC_s = \text{specific fuel consumption} \left(\frac{g_{eq\_dry\_fuel}}{L_{water}}\right)$$

$$\frac{1}{1000} = \text{conversion factor} \left(\frac{\text{kg}}{\text{g}}\right)$$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{CO2,s} = \frac{m_{CO2,s}}{w_{sr}} \cdot 1000$$

$E_{CO2,s}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$m_{CO2,s}$ = total emission (g)

w_{sr} = water remaining (g)

1000 = conversion factor $\left(\frac{g_{water}}{L_{water}}\right)$

26. Cold Start CO Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: W48

Old formula: W48=W52*W$33/1000

New formula: W48=W56/W28*1000

The old formula is expressed as: $E_{CO,c} = EF_{CO,c} \cdot SC_c \cdot \frac{1}{1000}$

$E_{CO,c}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$EF_{CO,c}$ = emission factor $\left(\frac{g_{emission}}{kg_{dry\_fuel}}\right)$

SC_c = specific fuel consumption $\left(\frac{g_{eq\_dry\_fuel}}{L_{water}}\right)$

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{CO,c} = \frac{m_{CO,c}}{w_{cr}} \cdot 1000$$

$E_{CO,c}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$m_{CO,c}$ = total emission (g)

w_{cr} = water boiled (g)

1000 = conversion factor $\left(\frac{g_{water}}{L_{water}}\right)$

27. Hot Start CO Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: AA48

Old formula: AA48=AA52*AA$33/1000

New formula: AA48=AA56/AA28*1000

The old formula is expressed as: $E_{CO,h} = EF_{CO,h} \cdot SC_h \cdot \frac{1}{1000}$

$E_{CO,h}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$EF_{CO,h}$ = emission factor $\left(\frac{g_{emission}}{kg_{dry\_fuel}}\right)$

SC_h = specific fuel consumption $\left(\frac{g_{eq\_dry\_fuel}}{L_{water}}\right)$

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{CO,h} = \frac{m_{CO,h}}{w_{hr}} \cdot 1000$$

$E_{CO,h}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$m_{CO,h}$ = total emission (g)

w_{hr} = water boiled (g)

$$1000 = \text{conversion factor}\ \left(\frac{g_{water}}{L_{water}}\right)$$

28. Simmer CO Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: AE48

Old formula: AE48=AE52*AP$32/1000

New formula: AE48=AE56/AP28*1000

The old formula is expressed as: $E_{CO,s} = EF_{CO,s} \cdot SC_s \cdot \frac{1}{1000}$

$$E_{CO,s} = \text{emission per water boiled}\ \left(\frac{g_{emission}}{L_{water}}\right)$$

$$EF_{CO,s} = \text{emission factor}\ \left(\frac{g_{emission}}{kg_{dry\_fuel}}\right)$$

$$SC_s = \text{specific fuel consumption}\ \left(\frac{g_{eq\_dry\_fuel}}{L_{water}}\right)$$

$$\frac{1}{1000} = \text{conversion factor}\ \left(\frac{kg}{g}\right)$$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{CO,s} = \frac{m_{CO,s}}{w_{sr}} \cdot 1000$$

$$E_{CO,s} = \text{emission per water boiled}\ \left(\frac{g_{emission}}{L_{water}}\right)$$

$$m_{CO,s} = \text{total emission (g)}$$

$$w_{sr} = \text{water remaining (g)}$$

$$1000 = \text{conversion factor}\ \left(\frac{g_{water}}{L_{water}}\right)$$

29. Cold Start PM Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: W49

Old formula: W49=W53*W$33/1000

New formula: W49=W57/W28*1000

The old formula is expressed as: $\mathrm{E_{PM,c}} = \mathrm{EF_{PM,c}} \cdot \mathrm{SC_c} \cdot \frac{1}{1000}$

$\mathrm{E_{PM,c}} = \text{emission per water boiled} \left(\frac{\mathrm{g_{emission}}}{\mathrm{L_{water}}}\right)$

$\mathrm{EF_{PM,c}} = \text{emission factor} \left(\frac{\mathrm{g_{emission}}}{\mathrm{kg_{dry\_fuel}}}\right)$

$\mathrm{SC_c} = \text{specific fuel consumption} \left(\frac{\mathrm{g_{eq\_dry\_fuel}}}{\mathrm{L_{water}}}\right)$

$\frac{1}{1000} = \text{conversion factor} \left(\frac{\mathrm{kg}}{\mathrm{g}}\right)$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$\mathrm{E_{PM,c}} = \frac{\mathrm{m_{PM,c}}}{\mathrm{w_{cr}}} \cdot 1000$$

$\mathrm{E_{PM,c}} = \text{emission per water boiled} \left(\frac{\mathrm{g_{emission}}}{\mathrm{L_{water}}}\right)$

$\mathrm{m_{PM,c}} = \text{total emission (g)}$

$\mathrm{w_{cr}} = \text{water boiled (g)}$

$1000 = \text{conversion factor} \left(\frac{\mathrm{g_{water}}}{\mathrm{L_{water}}}\right)$

30. Hot Start PM Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: AA49

Old formula: AA49=AA53*AA$33/1000

New formula: AA49=AA57/AA28*1000

The old formula is expressed as: $E_{PM,h} = EF_{PM,h} \cdot SC_h \cdot \frac{1}{1000}$

$E_{PM,h}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$EF_{PM,h}$ = emission factor $\left(\frac{g_{emission}}{kg_{dry\_fuel}}\right)$

SC_h = specific fuel consumption $\left(\frac{g_{eq\_dry\_fuel}}{L_{water}}\right)$

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{PM,h} = \frac{m_{PM,h}}{w_{hr}} \cdot 1000$$

$E_{PM,h}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$m_{PM,h}$ = total emission (g)

w_{hr} = water boiled (g)

1000 = conversion factor $\left(\frac{g_{water}}{L_{water}}\right)$

31. Simmer PM Emission per Water Boiled (g/L)

Sheets: Test-1,Test-2,Test-3

Cells: AE49

Old formula: AE49=AE53*AP$32/1000

New formula: AE49=AE57/AP28*1000

The old formula is expressed as: $E_{PM,s} = EF_{PM,s} \cdot SC_s \cdot \frac{1}{1000}$

$E_{PM,s}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$EF_{PM,s}$ = emission factor $\left(\frac{g_{emission}}{kg_{dry\_fuel}}\right)$

SC_s = specific fuel consumption $\left(\frac{g_{eq\_dry\_fuel}}{L_{water}}\right)$

$\frac{1}{1000}$ = conversion factor $\left(\frac{kg}{g}\right)$

The problem with this formula is that the equivalent dry fuel consumed in the specific consumption metric is not the same quantity as the dry fuel consumed in the emission factor metric. A better way to calculate the emission per water boiled is to divide the total emission [g] by the water boiled [L], resulting in the new formula:

$$E_{PM,s} = \frac{m_{PM,s}}{w_{sr}} \cdot 1000$$

$E_{PM,s}$ = emission per water boiled $\left(\frac{g_{emission}}{L_{water}}\right)$

$m_{PM,s}$ = total emission (g)

w_{sr} = water remaining (g)

1000 = conversion factor $\left(\frac{g_{water}}{L_{water}}\right)$

32. IWA Performance Metrics

Sheets: Test-1,Test-2,Test-3

Cells: AI39:AT47

A block of cells was inserted here to calculate and report IWA Performance Metrics and Tiers.

33. High Power Thermal Efficiency (%)

Sheets: Test-1,Test-2,Test-3

Cell: AS40

Formula: AS40=IF(AA29=0,W31,('Test-1'!W31+'Test-1'!AA31)/2)

If the Hot Start phase is omitted, then the cold start efficiency is reported:

$$\text{High Power Thermal Efficiency} = h_c$$

If the Hot Start phase is not omitted, then the average of the cold start efficiency and hot start efficiency is reported:

$$\text{High Power Thermal Efficiency} = \frac{h_c + h_h}{2}$$

h_c = Cold Start thermal efficiency (decimal fraction)

h_h = Hot Start thermal efficiency (decimal fraction)

The value of the metric is a decimal fraction but it is formatted as a percentage.

34. Low Power Specific Fuel Consumption (MJ/min/L)

Sheets: Test-1,Test-2,Test-3

Cell: AS41

Formula: AS41=AP26*'General Information'!L22/(AP28*AP29*1000)

$$\text{Low Power Specific Fuel Consumption} = \frac{f_{sd} \cdot \text{LHV}}{w_{sr} \cdot \Delta t_s \cdot 1000}$$

f_{sd} = equivalent dry wood consumed during simmer (g)

LHV = net calorific value of fuel (kJ/kg)

$w_{s,r}$ = water remaining at the end of the simmer period (g)

Δt_s = simmer time (minutes)

1000 = conversion factor (kJ/MJ)

35. High Power CO (g/MJ$_d$)

Sheets: Test-1,Test-2,Test-3

Cell: AS42

Formula: AS42=IF(AA29=0,1000000/'General Information'!L22*W56/(W31*W26),1000000/'General Information'!L22*(W56/(W31*W26)+AA56/(AA31*AA26))/2)

$$\text{High Power CO}\left(\frac{\text{g}}{\text{MJ}}\right) = \frac{\text{CO emission (g)}}{\text{Energy delivered to pot (MJ)}}$$

The energy delivered to the pot is the total energy consumed multiplied by the thermal efficiency.

If the Hot Start phase is omitted, then the metric is calculated for the Cold Start:

$$\text{High Power CO} = \frac{m_{CO,c} \cdot 1000000}{h_c \cdot f_{cd} \cdot LHV}$$

If the Hot Start phase is not omitted, then the metric is calculated for the Cold Start and Hot Start and the average of the two is reported:

$$\text{High Power CO} = \frac{\frac{m_{CO,c} \cdot 1000000}{h_c \cdot f_{cd} \cdot LHV} + \frac{m_{CO,h} \cdot 1000000}{h_h \cdot f_{hd} \cdot LHV}}{2} = \frac{1000000}{LHV \cdot 2} \cdot \left(\frac{m_{CO,c}}{h_c \cdot f_{cd}} + \frac{m_{CO,h}}{h_h \cdot f_{hd}}\right)$$

1000000 = conversion factor $\left(\frac{\text{kJ}\cdot\text{g}}{\text{MJ}\cdot\text{kg}}\right)$

LHV = net calorific value (kJ/kg)

$m_{CO,c}$ = CO mass during Cold Start (g)

$m_{CO,h}$ = CO mass during Hot Start (g)

h_c = Cold Start thermal efficiency

f_{cd} = Cold Start equivalent dry fuel consumed (g)

h_h = Hot Start thermal efficiency

f_{hd} = Hot Start equivalent dry fuel consumed (g)

36. Low Power CO (g/min/L)

Sheets: Test-1,Test-2,Test-3

Cell: AS43

Formula: AS43 =AE56/(AP29*AP28)*1000

$$\text{Low Power CO} = \frac{m_{CO,s}}{\Delta t_s \cdot w_{s,r}} \cdot 1000$$

$m_{CO,s}$ = total Simmer CO emissions (g)

Δt_s = Simmer time (minutes)

$w_{s,r}$ = water remaining at the end of the Simmer period (g)

1000 = conversion factor (g_{water}/L_{water})

37. High Power PM (mg/MJ$_d$)

Sheets: Test-1,Test-2,Test-3

Cell: AS44

Formula: AS44=IF(AA29=0,10^9/'General Information'!L22*W57/(W31*W26),10^9/'General Information'!L22*(W57/(W31*W26)+AA57/(AA31*AA26))/2)

$$High\ Power\ PM\left(\frac{\mathrm{mg}}{\mathrm{MJ}}\right) = \frac{PM\ emissions\ (mg)}{Energy\ delivered\ to\ pot\ (MJ)}$$

The energy delivered to the pot is the total energy consumed multiplied by the thermal efficiency.

If the Hot Start phase is omitted, then the metric is calculated for the Cold Start:

$$\text{High Power PM} = \frac{m_{PM,c} \cdot 10^9}{h_c \cdot f_{cd} \cdot LHV}$$

If the Hot Start phase is not omitted, then the metric is calculated for the Cold Start and Hot Start and the average of the two is reported:

$$\text{High Power PM} = \frac{\frac{m_{PM,c} \cdot 10^9}{h_c \cdot f_{cd} \cdot LHV} + \frac{m_{PM,h} \cdot 10^9}{h_h \cdot f_{hd} \cdot LHV}}{2} = \frac{10^9}{LHV \cdot 2} \cdot \left(\frac{m_{PM,c}}{h_c \cdot f_{cd}} + \frac{m_{PM,h}}{h_h \cdot f_{hd}}\right)$$

10^9 = conversion factor $\left(\frac{kJ \cdot g_{fuel} \cdot mg_{PM}}{MJ \cdot kg_{fuel} \cdot g_{PM}}\right)$

LHV = net calorific value (kJ/kg)

$m_{PM,c}$ = PM mass during Cold Start (g)

$m_{PM,h}$ = PM mass during Hot Start (g)

h_c = Cold Start thermal efficiency

f_{cd} = Cold Start equivalent dry fuel consumed (g)

h_h = Hot Start thermal efficiency

f_{hd} = Hot Start equivalent dry fuel consumed (g)

38. Low Power PM (mg/min/L)

Sheets: Test-1,Test-2,Test-3

Cell: AS45

Formula: AS45=AE57/(AP29*AP28)*1000000

$$\text{Low Power PM} = \frac{m_{PM,s}}{\Delta t_s \cdot w_{s,r}} \cdot 1000000$$

$m_{PM,s}$ = total Simmer PM emission (g)

Δt_s = Simmer time (minutes)

$w_{s,r}$ = water remaining at the end of the Simmer period (g)

1000000 = conversion factor$\left(\frac{g_{water} \cdot mg_{PM}}{L_{water} \cdot g_{PM}}\right)$

39. Indoor CO Emissions (g/min)

Sheets: Test-1,Test-2,Test-3

Cell: AS46

Formula: AS46=IF(IF(AA29=0,W56/W29,(W56/W29+AA56/AA29)/2)<(AE56/AP29),AE56/AP29, IF(AA29=0,W56/W29,(W56/W29+AA56/AA29)/2))

This metric reports the high power or low power CO emission rate into the kitchen, whichever is greater. If the Hot Start phase is omitted, then the high power emission rate is calculated for the Cold Start:

$$\text{High Power CO Emission Rate} = \frac{m_{CO,c}}{\Delta t_c}$$

If the Hot Start phase is not omitted, then the high power emission rate is calculated for the Cold Start and Hot Start and the average of the two is reported:

$$\text{High Power CO Emission Rate} = \frac{\frac{m_{CO,c}}{\Delta t_c} + \frac{m_{CO,h}}{\Delta t_h}}{2}$$

The low power emission rate is calculated from the simmer period:

$$\text{Low Power CO Emission Rate} = \frac{m_{CO,s}}{\Delta t_s}$$

$m_{CO,c}$ = Cold Start total CO emission (g)

$m_{CO,h}$ = Hot Start total CO emission (g)

$m_{CO,s}$ = Simmer total CO emission (g)

Δt_c = time of Cold Start (minutes)

Δt_h = time of Hot Start (minutes)

Δt_s = time of Simmer (minutes)

These formulas are only valid for non-chimney stoves that vent the total emissions into the kitchen.

40. Indoor PM Emissions (mg/min)

Sheets: Test-1,Test-2,Test-3

Cell: AS47

Formula: AS47=IF(IF(AA29=0,W57/W29,(W57/W29+AA57/AA29)/2)<(AE57/AP29), AE57/AP29*1000,IF(AA29=0,W57/W29,(W57/W29+AA57/AA29)/2)*1000)

This metric reports the high power or low power PM emission rate into the kitchen, whichever is greater. If the Hot Start phase is omitted, then the high power emission rate is calculated for the Cold Start:

$$\text{High Power PM Emission Rate} = \frac{m_{PM,c}}{\Delta t_c} \cdot 1000$$

If the Hot Start phase is not omitted, then the high power emission rate is calculated for the Cold Start and Hot Start and the average of the two is reported:

$$\text{High Power PM Emission Rate} = \frac{\left(\frac{m_{PM,c}}{\Delta t_c} + \frac{m_{PM,h}}{\Delta t_h}\right) \cdot 1000}{2}$$

The low power emission rate is calculated from the simmer period:

$$\text{Low Power PM Emission Rate} = \frac{m_{PM,s}}{\Delta t_s} \cdot 1000$$

$m_{PM,c} = \text{Cold Start total PM emission (g)}$

$m_{PM,h} = \text{Hot Start total PM emission (g)}$

$m_{PM,s} = \text{Simmer total PM emission (g)}$

$\Delta t_c = \text{ time of Cold Start (minutes)}$

$\Delta t_h = \text{ time of Hot Start (minutes)}$

$\Delta t_s = \text{ time of Simmer (minutes)}$

1000 = conversion factor (mg_{PM}/g_{PM})

These formulas are only valid for non-chimney stoves that vent the total emissions into the kitchen.

41. IWA Performance Tier Chart

Sheet: Lists

Cells: B14:M23

The IWA Tier Chart used to group the IWA performance metrics into performance tiers.

42. IWA Performance Tiers

Sheets: Test-1,Test-2,Test-3

Cells: AT39:AT47

Each IWA Performance Metric is binned into a performance tier according to the tier chart.

High Power Thermal Efficiency:
AT40=IF(AS40<Lists!E15,"0",IF(AS40<Lists!I15,1,IF(AS40<Lists!K15,2,IF(AS40<Lists!M15,3,4))))

Low Power Specific Fuel Consumption:
AT41=IF(AS41>Lists!E16,"0",IF(AS41>Lists!I16,1,IF(AS41>Lists!K16,2,IF(AS41>Lists!M16,3,4))))

High Power CO:
AT42=IF(AS42>Lists!E17,"0",IF(AS42>Lists!I17,1,IF(AS42>Lists!K17,2,IF(AS42>Lists!M17,3,4))))

Low Power CO:
AT43=IF(AS43>Lists!E18,"0",IF(AS43>Lists!I18,1,IF(AS43>Lists!K18,2,IF(AS43>Lists!M18,3,4))))

High Power PM:
AT44=IF(AS44>Lists!E19,"0",IF(AS44>Lists!I19,1,IF(AS44>Lists!K19,2,IF(AS44>Lists!M19,3,4))))

Low Power PM:
AT45=IF(AS45>Lists!E20,"0",IF(AS45>Lists!I20,1,IF(AS45>Lists!K20,2,IF(AS45>Lists!M20,3,4))))

Indoor CO Emissions:
AT46=IF(AS46>Lists!E21,"0",IF(AS46>Lists!I21,1,IF(AS46>Lists!K21,2,IF(AS46>Lists!M21,3,4))))

Indoor PM Emissions:
AT47=IF(AS47>Lists!E22,"0",IF(AS47>Lists!I22,1,IF(AS47>Lists!K22,2,IF(AS47>Lists!M22,3,4))))

43. Results Sheet Format Changes

Sheet: Results

Cells: D47:L65

IWA Performance Metrics and Tiers were added to the Results Sheet. For each performance metric, the average performance tier value (J57:J65) is the average value of that performance metric (J48:J55) binned into the tier chart. There remains space on the Results sheet (K57:K65) to calculate the precision of the IWA Metrics. There also remains space on the Results sheet (L57:L65) to perform a significance test to determine if the IWA performance metrics meet the precision requirement.

44. “Wood” changed to “fuel”

Sheets: Test Entry, Test-1, Test-2, Test-3

Cells: several

The word “wood” was replaced with the word “fuel” in several places.

45. Fuel Moisture Content Variable Label

Sheets: Test-1,Test-2,Test-3

Cell: I18

The label was changed from m to MC.

46. Effective Heating Value Variable Label

Sheets: Test-1,Test-2,Test-3

Cell: I22

The label was changed from c_{eff} to EHV.

47. Emission Report Variable Labels

Sheets: Test-1,Test-2,Test-3

Cells: X41:X61, AB41:AB57, AF41:AF57

Labels were added for all emissions variables.

48. Number Formats

Sheets: General Information, Test-1, Test-2, Test-3, Results

Cells: too many to count

The cell number formats were changed for many of the cells in the spreadsheet in order for the cells to display a meaningful output with an appropriate number of significant figures.

49. Ethanol Calorific Value (kJ/kg)

Sheet: Calorific Values

Cells: I6,K6

Old Values: HHV = 29,800

LHV = 26,700

New Values: HHV = 26,800

LHV = 24,200

The gross calorific value of ethanol was changed to match the value reported in Appendix 2 of the WBT pdf from the reference Schmer et al., 2007. The net calorific value was calculated by subtracting the latent heat of vaporization of water in the combustion products (assumed to be 2,600 kJ/kg).

Suggestions for Further Improvement:

The emission calculations could be further improved by adding an input of duct temperature for each phase of the test instead of one average flue temp value for all three phases.

Many cells in the spreadsheet display “DIV/0” or “VALUE” when important data is missing or entered incorrectly into the spreadsheet. Although this feature is a good way to check for errors, “if” statements can be added to each cell to show a blank cell instead, which will make the overall spreadsheet look cleaner and less confusing.

A consistent number format should be applied to the entire spreadsheet. Some cells display a negative number by using parenthesis while other cells put a negative sign in front of the number.

If the Hot Start phase is omitted, high power IWA Performance Metrics are reported based on the Cold Start. If it is desired that the IWA metrics only be applicable for full WBTs (Cold Start, Hot Start, and Simmer), then the high power IWA metric calculations should be changed so they report an error if the Hot Start is omitted.

The spreadsheet should be locked and protected.